
Projekt:  Werkstückanschlag 

Aufgabe:  Seitenteil Pos. 1; 2. Aufspannung 

Programmierung: Heidenhain  TNC 7  

Autor:
Klasse:
Datum:

Hauptprogramm
0  BEGIN PGM Wkstckanschl_H MM
1  BLK FORM 0.1 Z  X-80  Y-20  Z-30 
2  BLK FORM 0.2  X+80  Y+20  Z+1 
3  TOOL CALL 10 Z S8000 F500 
4  L  X+0  Y+0 R0 FMAX M3 
5  CALL PGM Abzeilen 
6  CALL PGM Rechteckzapf 
7  TOOL CALL 10 Z S8000 F500 DL+0.2 DR+0.5 
8  CALL PGM Kontur 
9  TOOL CALL 4 Z S10000 F600 DR+5 
10 L  X+0  Y-35 R0 FMAX M3 
11 L  Z+2 R0 FMAX 
12 L  Z-16.505 R0 F AUTO
13 CALL PGM Kontur1 
14 L  X+0  Y-35 
15 TOOL CALL 4 Z S10000 F600 DR+0.5 
16 CALL PGM Kontur1 
17 L  X+0  Y-35 
18 TOOL CALL 4 Z S10000 F600 
19 L  X+0  Y-35 R0 FMAX M3 
20 L  Z+2 R0 FMAX 
21 L  Z-16.505 R0 F AUTO
22 CALL PGM Kontur1 
23 L  Z+50 R0 FMAX 
24 CALL PGM Zyklen 
25 CALL PGM Zyklen2 
26 L  Z+100 R0 FMAX 
27 CALL PGM Entgraten 
28 TOOL CALL 0 Z 
29 M30 
30 END PGM Wkstckanschl_H MM

Programm Abzeilen
0  BEGIN PGM Abzeilen MM
1  L  X+0  Y+0 R0 FMAX M3 
2  CYCL DEF 230 ABZEILEN ~ 
    Q225=-90   ;STARTPUNKT 1. ACHSE ~ 
    Q226=-30   ;STARTPUNKT 2. ACHSE ~ 

    Q227=+0    ;STARTPUNKT 3. ACHSE ~ 
    Q218=+180  ;1. SEITEN-LAENGE ~ 
    Q219=+60   ;2. SEITEN-LAENGE ~ 
    Q240=+10   ;ANZAHL SCHNITTE ~ 
    Q206=+150  ;VORSCHUB TIEFENZ. ~ 
    Q207= AUTO ;VORSCHUB FRAESEN ~ 
    Q209=+150  ;VORSCHUB QUER ~ 
    Q200=+2    ;SICHERHEITS-ABST. 
3  CYCL CALL POS  X+0  Y+0  Z+0 
4  END PGM Abzeilen MM  

Programm Rechteckzapfen

0  BEGIN PGM Rechteckzapf MM  
1  CYCL DEF 256 RECHTECKZAPFEN ~ 
    Q218=+51   ;1. SEITEN-LAENGE ~ 
    Q424=+170  ;ROHTEILMASS 1 ~ 
    Q219=+30   ;2. SEITEN-LAENGE ~ 
    Q425=+45   ;ROHTEILMASS 2 ~ 
    Q220=+0    ;ECKENRADIUS ~ 
    Q368=+0    ;AUFMASS SEITE ~ 
    Q224=+0    ;DREHLAGE ~ 
    Q367=+0    ;ZAPFENLAGE ~ 
    Q207=+500  ;VORSCHUB FRAESEN ~ 
    Q351=+1    ;FRAESART ~ 
    Q201=-4.5  ;TIEFE ~ 
    Q202=+5    ;ZUSTELL-TIEFE ~ 
    Q206=+300  ;VORSCHUB TIEFENZ. ~ 
    Q200=+2    ;SICHERHEITS-ABST. ~ 
    Q203=+0    ;KOOR. OBERFLAECHE ~ 
    Q204=+50   ;2. SICHERHEITS-ABST. ~ 
    Q370=+0.8  ;BAHN-UEBERLAPPUNG 
2  CYCL CALL POS  X+0  Y+0  Z+0 
3  END PGM Rechteckzapf MM  

Programm Kontur

0  BEGIN PGM Kontur MM  
1  L  X+0  Y-35 R0 FMAX M3 
2  L  Z+2 R0 FMAX 
3  L  Z-6 F AUTO
4  M120 LA8 
5  CALL PGM Kontur1 
6  L  Z-11
7  M120 LA8 
8  CALL PGM Kontur1 
9  L  Z-16.505
10 M120 LA8 
11 CALL PGM Kontur1 
12 M120 LA8 
13 CALL PGM Kontur1 
14 END PGM Kontur MM  

Programm Kontur1 

0  BEGIN PGM Kontur1 MM  
1  LBL 1 
2  M120 LA10 

3  APPR LCT  X+0  Y-15 R2 RL F AUTO 
4  L  X-25.5
5  RND R5 
6  L  Y-3
7  RND R5 
8  L  X-45.303
9  RND R5 
10 CC  X-60  Y+0 
11 C  X-45.303  Y+3 DR- F AUTO
12 RND R5 
13 L  X-25.5
14 RND R5 
15 L  Y+15 
16 RND R5 
17 L  X-16.583
18 RND R5 
19 CC  X+0  Y+40 
20 C  X+16.583  Y+15 DR+ 
21 RND R5 
22 L  X+25.5 
23 RND R5 
24 L  Y+3 
25 RND R5 
26 L  X+45.303 
27 RND R5 
28 CC  X+60  Y+0 
29 C  X+45.303  Y-3 DR- 
30 RND R5 
31 L  X+25.5 
32 RND R5 
33 L  Y-15
34 RND R5 
35 L  X+16.583 
36 RND R5 
37 CC  X+0  Y-40 
38 C  X-16.583  Y-15 DR+
39 RND R5 
40 L  X-25.5
41 DEP LT LEN5 
42 L  X+0  Y-35 
43 LBL 0 
44 END PGM Kontur1 MM  

Programm Zyklen

0  BEGIN PGM Zyklen MM  
1  TOOL CALL 5 Z S7000 F500 
2  M3 
3  CYCL DEF 252 KREISTASCHE ~ 
    Q215=+0    ;BEARBEITUNGS-UMFANG ~ 
    Q223=+20   ;KREISDURCHMESSER ~ 
    Q368=+0.3  ;AUFMASS SEITE ~ 
    Q207=+500  ;VORSCHUB FRAESEN ~ 
    Q351=+1    ;FRAESART ~ 
    Q201=-12.05 ;TIEFE ~ 
    Q202=+3.1  ;ZUSTELL-TIEFE ~ 
    Q369=+0.2  ;AUFMASS TIEFE ~ 
    Q206=+150  ;VORSCHUB TIEFENZ. ~ 
    Q338=+0.5  ;ZUST. SCHLICHTEN ~ 
    Q200=+2    ;SICHERHEITS-ABST. ~ 
    Q203=-4.5  ;KOOR. OBERFLAECHE ~ 

    Q204=+50   ;2. SICHERHEITS-ABST. ~ 
    Q370=+0.8  ;BAHN-UEBERLAPPUNG ~ 
    Q366=+1    ;EINTAUCHEN ~ 
    Q385=+500  ;VORSCHUB SCHLICHTEN 
4  CYCL CALL POS  X-60  Y+0  Z+0 
5  CYCL CALL POS  X+60  Y+0  Z+0 
6  CYCL DEF 251 RECHTECKTASCHE ~ 
    Q215=+0    ;BEARBEITUNGS-UMFANG ~ 
    Q218=+30   ;1. SEITEN-LAENGE ~ 
    Q219=+15   ;2. SEITEN-LAENGE ~ 
    Q220=+5    ;ECKENRADIUS ~ 
    Q368=+0.3  ;AUFMASS SEITE ~ 
    Q224=+0    ;DREHLAGE ~ 
    Q367=+0    ;TASCHENLAGE ~ 
    Q207=+500  ;VORSCHUB FRAESEN ~ 
    Q351=+1    ;FRAESART ~ 
    Q201=-16.5 ;TIEFE ~ 
    Q202=+5.5  ;ZUSTELL-TIEFE ~ 
    Q369=+0.2  ;AUFMASS TIEFE ~ 
    Q206=+150  ;VORSCHUB TIEFENZ. ~ 
    Q338=+0.3  ;ZUST. SCHLICHTEN ~ 
    Q200=+2    ;SICHERHEITS-ABST. ~ 
    Q203=+0    ;KOOR. OBERFLAECHE ~ 
    Q204=+50   ;2. SICHERHEITS-ABST. ~ 
    Q370=+0.8  ;BAHN-UEBERLAPPUNG ~ 
    Q366=+1    ;EINTAUCHEN ~ 
    Q385=+500  ;VORSCHUB SCHLICHTEN 
7  CYCL CALL POS  X+0  Y+0  Z+0 
8  END PGM Zyklen MM  

Programm Zyklen2

0  BEGIN PGM Zyklen2 MM  
1  TOOL CALL 39 Z S5000 F400 
2  M3 
3  L  X-20  Y+0 R0 FMAX 
4  CYCL DEF 240 ZENTRIEREN ~ 
    Q200=+2    ;SICHERHEITS-ABST. ~ 
    Q343=+1    ;AUSWAHL DURCHM/TIEFE ~ 
    Q201=-2    ;TIEFE ~ 
    Q344=-6.3  ;DURCHMESSER ~ 
    Q206=+150  ;VORSCHUB TIEFENZ. ~ 
    Q211=+0.5  ;VERWEILZEIT UNTEN ~ 
    Q203=+0    ;KOOR. OBERFLAECHE ~ 
    Q204=+50   ;2. SICHERHEITS-ABST. 
5  CYCL CALL POS  X-20  Y+0  Z+0 
6  CYCL DEF 240 ZENTRIEREN ~ 
    Q200=+2    ;SICHERHEITS-ABST. ~ 
    Q343=+1    ;AUSWAHL DURCHM/TIEFE ~ 
    Q201=-2    ;TIEFE ~ 
    Q344=-6    ;DURCHMESSER ~ 
    Q206=+150  ;VORSCHUB TIEFENZ. ~ 
    Q211=+0.5  ;VERWEILZEIT UNTEN ~ 
    Q203=+0    ;KOOR. OBERFLAECHE ~ 
    Q204=+50   ;2. SICHERHEITS-ABST. 
7  CYCL CALL POS  X+20  Y+0  Z+0 
8  ; 
9  ;Werkzeug SPIBO D=4,2mm 
10 ;Werkzeugaufruf 
11 TOOL CALL 37 Z S2000 F500 
12 M3 

13 ; 
14 L  X+20  Y+0 R0 FMAX 
15 L  Z+2 R0 FMAX 
16 ; 
17 CYCL DEF 205 UNIVERSAL-TIEFBOHREN ~ 
    Q200=+2    ;SICHERHEITS-ABST. ~ 
    Q201=-18.5 ;TIEFE ~ 
    Q206=+150  ;VORSCHUB TIEFENZ. ~ 
    Q202=+5    ;ZUSTELL-TIEFE ~ 
    Q203=+0    ;KOOR. OBERFLAECHE ~ 
    Q204=+50   ;2. SICHERHEITS-ABST. ~ 
    Q212=+0    ;ABNAHMEBETRAG ~ 
    Q205=+0    ;MIN. ZUSTELL-TIEFE ~ 
    Q258=+0.2  ;VORHALTEABSTAND OBEN ~ 
    Q259=+0.2  ;VORHALTEABST. UNTEN ~ 
    Q257=+0    ;BOHRTIEFE SPANBRUCH ~ 
    Q256=+0.2  ;RZ BEI SPANBRUCH ~ 
    Q211=+0    ;VERWEILZEIT UNTEN ~ 
    Q379=+0    ;STARTPUNKT ~ 
    Q253=+750  ;VORSCHUB VORPOS. 
18 ; 
19 CYCL CALL POS  X+20  Y+0  Z+0 
20 ; 
21 ;Werkzeug GEWBO M5 
22 ;Werkzeugaufruf 
23 TOOL CALL 38 Z S200 
24 M3 
25 ; 
26 L  X+20  Y+0 R0 FMAX 
27 L  Z+2 R0 FMAX 
28 ; 
29 CYCL DEF 207 GEW.-BOHREN GS NEU ~ 
    Q200=+2    ;SICHERHEITS-ABST. ~ 
    Q201=-14.5 ;GEWINDETIEFE ~ 
    Q239=+0.8  ;GEWINDESTEIGUNG ~ 
    Q203=+0    ;KOOR. OBERFLAECHE ~ 
    Q204=+50   ;2. SICHERHEITS-ABST. 
30 ; 
31 CYCL CALL POS  X+20  Y+0  Z+0 
32 ; 
33 ;Werkzeug SPIBO D=5,5mm 
34 ;Werkzeugaufruf 
35 TOOL CALL 40 Z S4000 F400 
36 M3 
37 ; 
38 L  X-20  Y+0 R0 FMAX 
39 L  Z+2 R0 FMAX 
40 ; 
41 CYCL DEF 205 UNIVERSAL-TIEFBOHREN ~ 
    Q200=+2    ;SICHERHEITS-ABST. ~ 
    Q201=-24   ;TIEFE ~ 
    Q206=+150  ;VORSCHUB TIEFENZ. ~ 
    Q202=+6    ;ZUSTELL-TIEFE ~ 
    Q203=+0    ;KOOR. OBERFLAECHE ~ 
    Q204=+50   ;2. SICHERHEITS-ABST. ~ 
    Q212=+0    ;ABNAHMEBETRAG ~ 
    Q205=+0    ;MIN. ZUSTELL-TIEFE ~ 
    Q258=+0.2  ;VORHALTEABSTAND OBEN ~ 
    Q259=+0.2  ;VORHALTEABST. UNTEN ~ 
    Q257=+0    ;BOHRTIEFE SPANBRUCH ~ 
    Q256=+0.2  ;RZ BEI SPANBRUCH ~ 
    Q211=+0    ;VERWEILZEIT UNTEN ~ 
    Q379=+0    ;STARTPUNKT ~ 

    Q253=+750  ;VORSCHUB VORPOS. 
42 ; 
43 CYCL CALL POS  X-20  Y+0  Z+0 
44 ; 
45 L  Z+50 R0 FMAX 
46 END PGM Zyklen2 MM  

Programm Entgraten

0  BEGIN PGM Entgraten MM  
1  ; 
2  ;Werkzeug Fasenfraeser D=6mm 
3  ;Werkzeugaufruf 
4  TOOL CALL 41 Z S8000 F500 DR-0.9 
5  M3 
6  ; 
7  L  X-28  Y-10 R0 FMAX 
8  L  Z+2 R0 FMAX 
9  L  Z-7 R0 F AUTO 
10 ; 
11 ;Kontur 
12 APPR CT  X-30.5  Y-3 R+1.5 RL 
13 L  X-45.303
14 RND R5 
15 CC  X-60  Y+0 
16 C  X-45.303  Y+3 DR-
17 RND R5 
18 L  X-30.5
19 DEP LCT  X-35  Y+15 R1.5 
20 L  Z+2 
21 L  X+28  Y+10 R0 FMAX 
22 L  Z-7 R0 F AUTO
23 APPR LCT  X+30.5  Y+3 R1.5 RL 
24 L  X+45.303 
25 RND R5 
26 CC  X+60  Y+0 
27 C  X+45.303  Y-3 DR- 
28 RND R5 
29 L  X+30.5 
30 DEP LCT  X+35  Y-10 R1.5 
31 L  Z+2 R0 FMAX 
32 L  X-10  Y-25 R0 FMAX
33 L  Z-2.5 R0 F AUTO
34 APPR LT  X-15  Y-15 LEN5 RL 
35 L  X-25.5
36 RND R5 
37 L  Y+15 
38 RND R5 
39 L  X-16.583
40 RND R5 
41 CC  X+0  Y+40 
42 C  X+16.583  Y+15 DR+ 
43 RND R5 
44 L  X+25.5 
45 RND R5 
46 L  Y-15
47 RND R5 
48 L  X+16.583 
49 RND R5 
50 CC  X+0  Y-40 

51 C  X-16.583  Y-15 DR+
52 RND R5 
53 L  X-25.5
54 DEP LT LEN5 
55 ; 
56 L  Z+2 R0 FMAX 
57 L  X+0  Y+0 R0 FMAX 
58 L  Z-2.5 R0 F AUTO
59 ; 
60 APPR LCT  X+0  Y+7.5 R2 RL 
61 L  X-15
62 RND R5 
63 L  Y-7.5
64 RND R5 
65 L  X+15 
66 RND R5 
67 L  Y+7.5 
68 RND R5 
69 L  X-1
70 DEP LCT  X+0  Y+0 R2 
71 L  Z+2 R0 FMAX 
72 ; 
73 L  X-60  Y+0 R0 FMAX
74 L  Z-7 R0 F AUTO
75 ; 
76 APPR LCT  X-50  Y+0 R3 RL 
77 CC  X-60  Y+0 
78 C  X-50  Y+0 DR+
79 DEP LCT  X-60  Y+0 R3 
80 ; 
81 L  Z+2 R0 FMAX 
82 L  X+60  Y+0 R0 FMAX 
83 L  Z-7 R0 F AUTO
84 APPR LCT  X+70  Y+0 R3 RL 
85 CC  X+60  Y+0 
86 C  X+70  Y+0 DR+ 
87 DEP LCT  X+60  Y+0 R3 
88 L  Z+50 R0 FMAX 
89 ; 
90 TOOL CALL 0 Z 
91 ; 
92 M30 
93 END PGM Entgraten MM  

